
%--------------------------------------------------------------------------------------------------- 
%  zj(+Vstup, - Vystup)   
% zjednodušuje výrazy, které jsou součtem členů tvaru  +- K*A , 
% kde K je číslo a A term. Např. 
% 2*(a+b) + 5 -3*a + 5*b - 4*(a+b) – 2 + 18*b    na 
% -2*(a+b) - 3*a + 23*b + 3 
%--------------------------------------------------------------------------------------------------- 
 
zj(Vstup,Vystup):-zprac(Vstup,0,[],Vystup). 
 
%--------------------------------------------------------------------------------------------------- 
% zprac(+Vstup, Aritm_akum, Term_akum, - Vystup). 
%  Ze zpracování vstupního výrazu Vstup a 
%  naakumulovaných hodnot v akumulátorech  
%  se spočte výsledný výraz Vystup.  
%  Akumulátor Aritm_akum je číslo, ve kterém se udržuje 
%    aktuální součet absolutních členů 
%  Akumulátor Term_akum je seznam dvojic typu a(K,A), 
%    které reprezentují akumulované nenulové "K-násobky termu A" 
%--------------------------------------------------------------------------------------------------- 
 
% výraz +- číslo 
zprac(V+A,N,Scit,Vyst):- number(A), ! , N1 is N+A, 
      zprac(V,N1,Scit,Vyst). 
zprac(V-A,N,Scit,Vyst):- number(A), ! , N1 is N-A, 
      zprac(V,N1,Scit,Vyst). 
 
% výraz +- číslo * něco 
zprac(V+K*A,N,Scit,Vyst):- number(K),! , 
        pridej(K,A,Scit,Scit1), 
        zprac(V,N,Scit1,Vyst). 
zprac(V-K*A,N,Scit,Vyst):- number(K), ! , K1 is -K, 
        pridej(K1,A,Scit,Scit1), 
        zprac(V,N,Scit1,Vyst). 
 
% výraz +- něco 

 
zprac(V+A,N,Scit,Vyst):- pridej(1,A,Scit,Scit1), 
       zprac(V,N,Scit1,Vyst). 
zprac(V-A,N,Scit,Vyst):- pridej(-1,A,Scit,Scit1), 
       zprac(V,N,Scit1,Vyst). 
 
% jediný člen číslo 
zprac(A,N,[],N1):-     number(A),!,N1 is N+A. 
zprac(A,N,Scit,Vyst):- number(A),!, N1 is N+A, 
     seber(N1,Scit,Vyst). 
 



 
%jediný člen tvaru +- K*A 
zprac(K*A,N,Scit,Vyst):-number(K), ! , 

pridej(K,A,Scit,Scit1), 
     seber(N,Scit1,Vyst). 
zprac(-K*A,N,Scit,Vyst):-number(K), ! , K1 is –K, 

  pridej(K1,A,Scit,Scit1), 
       seber(N,Scit1,Vyst). 
 
%jediný člen v ostatních případech 
zprac(A,N,Scit,Vyst):- pridej(1,A,Scit,Scit1), 
     seber(N,Scit1,Vyst). 
zprac(-A,N,Scit,Vyst):- pridej(-1,A,Scit,Scit1), 
     seber(N,Scit1,Vyst). 
 
 
%--------------------------------------------------------------------------------------------------- 
%  pridej(+K,+A,+OldAc,-NewAc) 
%  přidá K-násobek A do akumulátoru OldAc a vytvoří tak nový akumulátor NewAc 
%--------------------------------------------------------------------------------------------------- 
 
% přidávám K-násobek termu A, který je v hlavě akumulátoru termů 
pridej(K,A,[a(L,A)|Z],Z):-  0 =:= L+K , ! . 
pridej(K,A,[a(L,A)|Z],[a(L1,A)|Z]):- ! , L1 is L+K . 
 
% přidávám K-násobek termu A, který není v hlavě akumulátoru termů 
pridej(K,A,[X|Z],[X|Z1]):- pridej(K,A,Z,Z1). 
 
% přidávám K-násobek termu A do prázdného akumulátoru termů 
pridej(K,A,[],[a(K,A)]):- K=\=0. 
pridej(K,_,[],[]):- K=:=0. 
 



 
%--------------------------------------------------------------------------------------------------- 
% seber(+Aritm_akum, +Term_akum, - Vystup) 
% Vytvoří z aritmetického akumulátoru a akumulátoru termů 
% výsledný Výraz 
%--------------------------------------------------------------------------------------------------- 
 
seber(N,[],N) :- ! . 
seber(0,Scit,Vyst):- ! , seber0(Scit,Vyst). 
seber(N,Scit,Vyst+N):- seber0(Scit,Vyst). 
 
%--------------------------------------------------------------------------------------------------- 
% seber0(+Term_akum, - Vystup). 
% Vytvoří z akumulátoru termů výsledný Výraz 
%--------------------------------------------------------------------------------------------------- 
 
seber0([a(K,A)],K*A). 
seber0([a(K,A)|Z],Vyst+K*A):- K>0,!, 
    seber0(Z,Vyst). 
seber0([a(K,A)|Z],Vyst-K1*A):- K1 is -K, 
    seber0(Z,Vyst). 


